NAG Library Routine Document

F08QLF (DTRSNA)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08QLF (DTRSNA) estimates condition numbers for specified eigenvalues and/or right eigenvectors of a real upper quasi-triangular matrix.

2 Specification

```
 SUBROUTINE F08QLF(JOB, HOWMNY, SELECT, N, T, LDT, VL, LDVL, VR, LDVR, S,
1                  SEP, MM, M, WORK, LDWORK, IWORK, INFO)
 INTEGER           N, LDT, LDVL, LDVR, MM, M, LDWORK, IWORK(*), INFO
 double precision  T(LDT,*), VL(LDVL,*), VR(LDVR,*), S(*), SEP(*),
1                  WORK(LDWORK,*)
 LOGICAL           SELECT(*)
 CHARACTER*1       JOB, HOWMNY
```

The routine may be called by its LAPACK name ***dtrsna***.

3 Description

F08QLF (DTRSNA) estimates condition numbers for specified eigenvalues and/or right eigenvectors of a real upper quasi-triangular matrix T in canonical Schur form. These are the same as the condition numbers of the eigenvalues and right eigenvectors of an original matrix $A = ZTZ^{\mathrm{T}}$ (with orthogonal Z), from which T may have been derived.

F08QLF (DTRSNA) computes the reciprocal of the condition number of an eigenvalue λ_i as

$$s_i = \frac{\left|v^{\mathrm{H}}u\right|}{\|u\|_E\|v\|_E},$$

where u and v are the right and left eigenvectors of T, respectively, corresponding to λ_i. This reciprocal condition number always lies between zero (i.e., ill-conditioned) and one (i.e., well-conditioned).

An approximate error estimate for a computed eigenvalue λ_i is then given by

$$\frac{\epsilon\|T\|}{s_i},$$

where ϵ is the ***machine precision***.

To estimate the reciprocal of the condition number of the right eigenvector corresponding to λ_i, the routine first calls F08QFF (DTREXC) to reorder the eigenvalues so that λ_i is in the leading position:

$$T = Q\begin{pmatrix}\lambda_i & c^{\mathrm{T}}\\ 0 & T_{22}\end{pmatrix}Q^{\mathrm{T}}.$$

The reciprocal condition number of the eigenvector is then estimated as sep_i, the smallest singular value of the matrix $(T_{22} - \lambda_i I)$. This number ranges from zero (i.e., ill-conditioned) to very large (i.e., well-conditioned).

An approximate error estimate for a computed right eigenvector u corresponding to λ_i is then given by

$$\frac{\epsilon\|T\|}{sep_i}.$$

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: JOB – CHARACTER*1 *Input*

On entry: indicates whether condition numbers are required for eigenvalues and/or eigenvectors.

JOB = 'E'
Condition numbers for eigenvalues only are computed.

JOB = 'V'
Condition numbers for eigenvectors only are computed.

JOB = 'B'
Condition numbers for both eigenvalues and eigenvectors are computed.

Constraint: JOB = 'E', 'V' or 'B'.

2: HOWMNY – CHARACTER*1 *Input*

On entry: indicates how many condition numbers are to be computed.

HOWMNY = 'A'
Condition numbers for all eigenpairs are computed.

HOWMNY = 'S'
Condition numbers for selected eigenpairs (as specified by SELECT) are computed.

Constraint: HOWMNY = 'A' or 'S'.

3: SELECT($*$) – LOGICAL array *Input*

Note: the dimension of the array SELECT must be at least $\max(1, \mathrm{N})$ if HOWMNY = 'S', and at least 1 otherwise.

On entry: specifies the eigenpairs for which condition numbers are to be computed if HOWMNY = 'S'. To select condition numbers for the eigenpair corresponding to the real eigenvalue λ_j, SELECT(j) must be set .TRUE.. To select condition numbers corresponding to a complex conjugate pair of eigenvalues λ_j and λ_{j+1}, SELECT(j) and/or SELECT($j+1$) must be set to .TRUE..

If HOWMNY = 'A', SELECT is not referenced.

4: N – INTEGER *Input*

On entry: n, the order of the matrix T.

Constraint: N ≥ 0.

5: T(LDT,$*$) – ***double precision*** array *Input*

Note: the second dimension of the array T must be at least $\max(1, \mathrm{N})$.

On entry: the n by n upper quasi-triangular matrix T in canonical Schur form, as returned by F08PEF (DHSEQR).

6: LDT – INTEGER *Input*

On entry: the first dimension of the array T as declared in the (sub)program from which F08QLF (DTRSNA) is called.

Constraint: LDT $\geq \max(1, \mathrm{N})$.

7: VL(LDVL,$*$) – ***double precision*** array *Input*

Note: the second dimension of the array VL must be at least $\max(1, \mathrm{MM})$ if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.

On entry: if JOB = 'E' or 'B', VL must contain the left eigenvectors of T (or of any matrix QTQ^{T} with Q orthogonal) corresponding to the eigenpairs specified by HOWMNY and SELECT. The eigenvectors **must** be stored in consecutive columns of VL, as returned by F08PKF (DHSEIN) or F08QKF (DTREVC).

If JOB = 'V', VL is not referenced.

8: LDVL – INTEGER *Input*

On entry: the first dimension of the array VL as declared in the (sub)program from which F08QLF (DTRSNA) is called.

Constraints:

if JOB = 'E' or 'B', LDVL $\geq \max(1, \mathrm{N})$;
if JOB = 'V', LDVL ≥ 1.

9: VR(LDVR,$*$) – ***double precision*** array *Input*

Note: the second dimension of the array VR must be at least $\max(1, \mathrm{MM})$ if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.

On entry: if JOB = 'E' or 'B', VR must contain the right eigenvectors of T (or of any matrix QTQ^{T} with Q orthogonal) corresponding to the eigenpairs specified by HOWMNY and SELECT. The eigenvectors **must** be stored in consecutive columns of VR, as returned by F08PKF (DHSEIN) or F08QKF (DTREVC).

If JOB = 'V', VR is not referenced.

10: LDVR – INTEGER *Input*

On entry: the first dimension of the array VR as declared in the (sub)program from which F08QLF (DTRSNA) is called.

Constraints:

if JOB = 'E' or 'B', LDVR $\geq \max(1, \mathrm{N})$;
if JOB = 'V', LDVR ≥ 1.

11: S($*$) – ***double precision*** array *Output*

Note: the dimension of the array S must be at least $\max(1, \mathrm{MM})$ if JOB = 'E' or 'B' and at least 1 if JOB = 'V'.

On exit: the reciprocal condition numbers of the selected eigenvalues if JOB = 'E' or 'B', stored in consecutive elements of the array. Thus S(j), SEP(j) and the jth columns of VL and VR all correspond to the same eigenpair (but not in general the jth eigenpair unless all eigenpairs have been selected). For a complex conjugate pair of eigenvalues, two consecutive elements of S are set to the same value.

S is not referenced if JOB = 'V'.

12: SEP($*$) – ***double precision*** array *Output*

Note: the dimension of the array SEP must be at least $\max(1, \mathrm{MM})$ if JOB = 'V' or 'B' and at least 1 if JOB = 'E'.

On exit: the estimated reciprocal condition numbers of the selected right eigenvectors if JOB = 'V' or 'B', stored in consecutive elements of the array. For a complex eigenvector, two consecutive elements of SEP are set to the same value. If the eigenvalues cannot be reordered to compute SEP(j), then SEP(j) is set to zero; this can only occur when the true value would be very small anyway.

If JOB = 'E', SEP is not referenced.

13: MM – INTEGER *Input*

On entry: the number of elements in the arrays S and SEP, and the number of columns in the arrays VL and VR (if used). The precise number required, m, is n if HOWMNY = 'A'; if HOWMNY = 'S', m is obtained by counting 1 for each selected real eigenvalue, and 2 for each selected complex conjugate pair of eigenvalues (see SELECT), in which case $0 \leq m \leq n$.

Constraint: MM $\geq$ M.

14: M – INTEGER *Output*

On exit: m, the number of elements of S and/or SEP actually used to store the estimated condition numbers. If HOWMNY = 'A', M is set to n.

15: WORK(LDWORK,$*$) – ***double precision*** array *Workspace*

Note: the second dimension of the array WORK must be at least $\max(1, \mathrm{N}+6)$ if JOB = 'V' or 'B' and at least 1 if JOB = 'E'.

If JOB = 'E', WORK is not referenced.

16: LDWORK – INTEGER *Input*

On entry: the first dimension of the array WORK as declared in the (sub)program from which F08QLF (DTRSNA) is called.

Constraints:

if JOB = 'V' or 'B', LDWORK $\geq \max(1, \mathrm{N})$;
if JOB = 'E', LDWORK ≥ 1.

17: IWORK($*$) – INTEGER array *Workspace*

Note: the dimension of the array IWORK must be at least $\max(1, 2 \times (\mathrm{N}-1))$.

18: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed values sep_i may over estimate the true value, but seldom by a factor of more than 3.

8 Further Comments

For a description of canonical Schur form, see the document for F08PEF (DHSEQR).

The complex analogue of this routine is F08QYF (ZTRSNA).

9 Example

This example computes approximate error estimates for all the eigenvalues and right eigenvectors of the matrix T, where

$$T = \begin{pmatrix} 0.7995 & -0.1144 & 0.0060 & 0.0336 \\ 0.0000 & -0.0994 & 0.2478 & 0.3474 \\ 0.0000 & -0.6483 & -0.0994 & 0.2026 \\ 0.0000 & 0.0000 & 0.0000 & -0.1007 \end{pmatrix}.$$

9.1 Program Text

```
*     F08QLF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=8)
      INTEGER          LDT, LDVL, LDVR, LDWORK
      PARAMETER        (LDT=NMAX,LDVL=NMAX,LDVR=NMAX,LDWORK=NMAX)
*     .. Local Scalars ..
      DOUBLE PRECISION EPS, TNORM
      INTEGER          I, INFO, J, M, N
*     .. Local Arrays ..
      DOUBLE PRECISION S(NMAX), SEP(NMAX), T(LDT,NMAX), VL(LDVL,NMAX),
     +                 VR(LDVR,NMAX), WORK(LDWORK,NMAX+6)
      INTEGER          IWORK(2*NMAX-1)
      LOGICAL          SELECT(1)
*     .. External Functions ..
      DOUBLE PRECISION F06RAF, X02AJF
      EXTERNAL         F06RAF, X02AJF
*     .. External Subroutines ..
      EXTERNAL         DTREVC, DTRSNA
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08QLF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read T from data file
*
         READ (NIN,*) ((T(I,J),J=1,N),I=1,N)
*
*        Calculate the left and right eigenvectors of T
*
         CALL DTREVC('Both','All',SELECT,N,T,LDT,VL,LDVL,VR,LDVR,N,M,
     +               WORK,INFO)
*
*        Estimate condition numbers for all the eigenvalues and right
*        eigenvectors of T
*
         CALL DTRSNA('Both','All',SELECT,N,T,LDT,VL,LDVL,VR,LDVR,S,SEP,
     +               N,M,WORK,LDWORK,IWORK,INFO)
*
*        Print condition numbers of eigenvalues and right eigenvectors
*
         WRITE (NOUT,*) 'S'
         WRITE (NOUT,99999) (S(I),I=1,M)
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'SEP'
         WRITE (NOUT,99999) (SEP(I),I=1,M)
*
*        Calculate approximate error estimates (using the 1-norm)
*
         EPS = X02AJF()
         TNORM = F06RAF('1-norm',N,N,T,LDT,WORK)
         WRITE (NOUT,*)
```

```
         WRITE (NOUT,*) 'Approximate error estimates for eigenvalues ',
     +     'of T (machine-dependent)'
         WRITE (NOUT,99999) (EPS*TNORM/S(I),I=1,M)
         WRITE (NOUT,*)
         WRITE (NOUT,*) 'Approximate error estimates for right ',
     +     'eigenvectors of T (machine-dependent)'
         WRITE (NOUT,99999) (EPS*TNORM/SEP(I),I=1,M)
       ELSE
         WRITE (NOUT,*) 'NMAX too small'
       END IF
*
99999 FORMAT ((3X,1P,7E11.1))
      END
```

9.2 Program Data

```
F08QLF Example Program Data
  4                                    :Value of N
  0.7995  -0.1144   0.0060   0.0336
  0.0000  -0.0994   0.2478   0.3474
  0.0000  -0.6483  -0.0994   0.2026
  0.0000   0.0000   0.0000  -0.1007   :End of matrix T
```

9.3 Program Results

```
 F08QLF Example Program Results

 S
       9.9E-01    7.0E-01    7.0E-01    5.7E-01

 SEP
       6.3E-01    3.7E-01    3.7E-01    3.1E-01

 Approximate error estimates for eigenvalues of T (machine-dependent)
       9.6E-17    1.4E-16    1.4E-16    1.7E-16

 Approximate error estimates for right eigenvectors of T (machine-dependent)
       1.5E-16    2.6E-16    2.6E-16    3.1E-16
```